

Advanced Clinical Graphs using Axis Tables

Sanjay Matange, SAS Institute Inc.

ABSTRACT

A key feature of graphs used for the analysis data or for clinical research is the inclusion of textual data in the graph, usually aligned with the x or y axis. The Axis Table statements available with the SGPLOT procedure make it easier to add such data to the graphs. Axis tables can also be used for creating custom axes and multiple axes.

INTRODUCTION

Analysis of the data for clinical research and many other domains is made easier by presenting the data in graphical form. This makes it easier to compare responses for different treatments or visualize the laboratory results over time. Such graphs often require the display of data in textual form aligned with a numerical or discrete data on the x or y axis. A popular example is the Survival Plot shown in Figure 1.




Figure 1 - Survival Plot.

The survival plot displays the survival probabilities over time by different groups, in this case, the type of Leukemia. An important feature of the graph is the display of the number of patients in the study by cancer type at the bottom of the graph. The numerical values are displayed at the corresponding locations along the time axis representing the number of subjects in the study at that time.

Traditionally, such information was inserted into the graph as a post process, after the creation of the plot. Some space is reserved at the bottom of the graph by the user and then an “annotation” process is used to display the tabular data below the legend. This requires the programmer to estimate the space needed for the table, and reserve this space. Then the user has to code annotation instructions for each text string to be displayed in the reserved space, including the position of the string along the x-axis and its position from the bottom of the graph. When the information is classified by type, the values need to be displayed as a stack of multiple rows, with the class value for each row displayed on the left of the y-axis.

In addition to scripting the instructions needed to display the values, this process does not scale well to changing data, as the space that needs to be reserved can change. This process becomes even more tenuous when adding textual data columns to the left or right of the data as we do for a forest plot.

With SAS 9.4, the new AXISTABLE statement was introduced in GTL to make such graphs easier to create. This new statement was specifically designed to support the need to insert textual data aligned with numeric or discrete x or y axes. Similarly, XAXISTABLE and YAXISTABLE statements were introduced in the SGPLOT and SGPanel procedures for the same purpose. This paper will describe the key features of these new statements and how you can use them to create popular graphs like the survival plot or the forest plot.

In addition to creating these popular graphs, you can also use these features creatively to customize your graphs in other interesting ways. This paper will provide some such examples to show you other ways in which these statements. In this paper, we will discuss examples using the SGPLOT version of the statement for ease of use. You can also use this with GTL template based graphs.

THE AXISTABLE STATEMENT SYNTAX

The syntax for usage of the axis table statements is shown below. You can use the XAXISTABLE to place rows of textual data aligned with the x or x2 axis as in the survival plot shown in Figure 1. You can use the YAXISTABLE to place columns of textual data aligned with the y or y2 axis as in the forest plot shown later in this paper.

```
xaxistable variable <variable> / <x=variable> <options>;
yaxistable variable <variable> / <y=variable> <options>;
```

Normally, the XAXISTABLE or YAXISTABLE statement is used in conjunction with plot statements such as a SCATTER or VBAR or HBAR. The variable assigned to the primary axis role of such statements (such as X or CATEGORY) is used to align the values from the variables in the axis table statement(s). One or more variables can be specified to create stacks of rows or columns. Or, the table can be classified by a variable to display multiple rows or columns. You can also align the values on the axis using a different variable (of the same type), using the optional X= or Y= parameters as we will do in the survival plot example in Figure 5 and 6.

The axis table statements support a large number of data and appearance options. Some of the common options are listed below. Refer to the product documentation for all the details.

CLASS=var	Multiple rows or columns of data are displayed, one per unique value.
COLORGROUP=var	Text color is based on the unique values of this variable.
INDENTWEIGHT=num-var	Values are indented based on the value of this variable.
TEXTGROUP=var	Text attributes are displayed based on the unique values of this variable.

Table 2 - Data options

CLASSDISPLAY	The values per class level are displayed stacked or side by side.
INDENT=value	Specifies the indent value to be used as multiplier with INDENTWEIGHT
TEXTGROUPID=id	Attribute id in associated attribute map for text attributes.
LOCATION	Location inside or outside the plot area.
POSITION	Position of the row (above or below), or column (left or right) of the plot area
LABEL NOLABEL	Display or hide the variable name or label.
STAT	Aggregation statistics for repeated values for a specific category / group value

Table 3 - Appearance options

The INDENT and INDENTWEIGHT options can be used to indent the values in a column for the YAXISTABLE. TEXTGROUP and TEXTGROUPID with an attribute map can be used to apply different text attributes for the subgroups and values. These are useful when creating a subgrouped forest plot.

USING THE XAXISTABLE FOR SURVIVAL PLOT

Creating the popular survival plot becomes easier using the new XAXISTABLE statement. It is worth stating that a Survival Plot is created by the LIFETEST procedure. However, many users have expressed the desire to create a customized version. To do this, one can use the SGPLOT procedure including the new XAXISTABLE statement to display the table of subjects at risk.

To get started, I have extracted the data needed for the graph by running the LIFETEST procedure using the BMT data set. The code for the LIFETEST procedure step is shown in Figure 4.

```
ods output Survivalplot=data.SurvivalPlotData;
proc lifetest data=sashelp.BMT plots=survival(atrisk=0 to 2500 by 500);
  time T * Status(0);
  strata Group / test=logrank adjust=sidak;
run;
```

Figure 4 – Creating the data set for the graph

A snippet of the data set is displayed in Figure 5.

Obs	Time	Survival	AtRisk	Event	Censored	tAtRisk	Stratum	StratumNum
1	0	1.00000	38	0	.	.	1: ALL	1
2	0	.	38	.	.	0	1: ALL	1
3	1	0.97368	38	1	.	.	1: ALL	1
4	55	0.94737	37	1	.	.	1: ALL	1
5	74	0.92105	36	1	.	.	1: ALL	1

Figure 5 - Data for Survival plot.

The SGPLOT procedure code needed to create the basic Survival Plot is shown in Figure 6 and the resulting graph is shown in Figure 7 below.

```

title 'Product-Limit Survival Estimates';
title2 h=0.8 'With Number of Subjects at Risk';
proc sgplot data=data.SurvivalPlotData;
  step x=time y=survival / group=stratum name='s';
  scatter x=time y=censored / markerattrs=(symbol=plus) name='c';
  scatter x=time y=censored / markerattrs=(symbol=plus) GROUP=stratum;
  xaxistable atrisk / x=tatrisk class=stratum colorgroup=stratum;
  keylegend 'c' / location=inside position=topright;
  keylegend 's' / linelength=20;
run;

```

Figure 6 – Creating the survival plot




Figure 7 - Survival Plot with outer table of subjects at risk.

Note the following aspects of this program and graph.

- The STEP plot is used to display the survival probabilities by type over time.
- The XAXISTABLE statement is used to display the table of subjects at risk at the bottom.
- The default alignment variable for the axis table would be the "time" variable used with the STEP plot.

- In this case, the option `X=tatrisk` is used to align the values with this variable.
- The values are displayed from the “atrisk” column, classified by `CLASS=stratum`.
- Since the “stratum” variable has three unique values, three rows of values are displayed.
- Each row displays the value of the class variable on the left.
- The values and the class values in the table are colored using the `COLORGROUP` option.
- The appropriate amount of space is allocated to display the values and rows.
- The appropriate amount of space is allocated to the left of the graph for the row headers.
- The graph will adjust automatically for changing number of class values or string lengths.
- Note, the placement of the y-axis label farther away from the y-axis values. This is due to the placement of the table in a separate cell below the graph. See Figure 7 for an alternative layout.
- No annotation is needed to create this graph.

CUSTOMIZING THE SURVIVAL PLOT

Recently, a user called asking about the extra whitespace on the left of the first tick value on the x-axis, as can be seen in Figure 7. It was important for the user to have the first tick and value be to the extreme left, in line with the y-axis line, and for the survival curves start from touching the y-axis.

Note, the reason for the extra white space is to prevent the first tick value being clipped. Also, when the “Subjects at Risk” table is added at the bottom, space is needed to avoid clipping the first column of the values. Setting the x-axis `OFFSETMIN=0` would remove the extra space, but it would also result in clipping of the first tick value and the first column of data in the “Subjects at Risk” table below.

The graph shown in Figure 8 below addresses this issue without clipping the first values. To do this, I did not set `OFFSETMIN=0`, as that would clip the values. Instead, I suppressed the y-axis ticks and line and set `AXISEXTENT=data`. I added a `REFLINE` at `x=0`, and a few `DROPLINES` for the ticks. This gives the appearance that the user wants.

Also note in Figure 8, I added `CURVELABELS` to display the stratum values at the ends of the curves, using splits. This places the information right next to the line, and obviates the need for the legend, which is removed. This creates an alternate view with reduced clutter. This graph can be displayed in monochrome output without loss of information, as shown in Figure 8.




Figure 8 - Survival Plot with inner table of subjects at risk.

USING THE YAXISTABLE FOR FOREST PLOT

Another popular graph used for clinical research is the forest plot. In this example, the requirement is to display the study data by various subgroups using the data shown in Figure 9.

ObsId	Id	Subgroup	indentWt	countpct	Low	Mean	High	PCIGroup	Group	PValue	ref
1	1	Overall	0	2166(100)	0.90	1.3	1.50	17.2	15.6	.	.
2	1	Age	0		0.05	2
3	2	<= 65 Yr	1	1534(71)	1.05	1.5	1.90	17.0	13.2	.	3
4	2	> 65 Yr	1	632(29)	0.60	0.8	1.25	17.8	21.3	.	4
5	1	Sex	0		0.13	.

Figure 9 - Data for forest plot.

The program to create this graph is shown in Figure 10, and the graph is shown in Figure 11.

```
proc sgplot data=forest_subgroup_2 nowall noborder nocycleattrs
  datrmap=attrmap noautolegend;
  styleattrs axisextent=data;
  reffline ref / lineattrs=(color=cxe7e7f7) discretethickness=1;
  highlow y=obsid low=low high=high;
  scatter y=obsid x=mean / markerattrs=(symbol=squarefilled size=4);
  scatter y=obsid x=mean / markerattrs=(size=0) x2axis;
  reffline 1 / axis=x;
  text x=x1 y=obsid text=text1 / position=left contributeoffsets=none;
  text x=x1 y=obsid text=text2 / position=right contributeoffsets=none;
  yaxistable subgroup / location=inside position=left textgroup=id
    textgroupid=text indentweight=indentWt;
  yaxistable countpct / location=inside position=left
  yaxistable PCIGroup group pvalue / location=inside position=right;
  yaxis reverse display=none type=discrete
    colorbands=odd colorbandsattrs=(transparency=1);
  xaxis display=(nolabel) values=(0.0 0.5 1.0 1.5 2.0 2.5);
  x2axis label='Hazard Ratio' display=(noline noticks novalues);
run;
```

Figure 10 – SGPLOT procedure code for the forest plot




Figure 11 – Forest plot

Note the following aspects of this program and graph.

- A HIGHLOW plot and a SCATTER plot are used to display the hazard ratios in the middle.
- One YAXISTABLE statement is used to display the sub grouped study names on the left.
 - The TEXTGROUP and TEXTGROUPID options are used for the bold font for the subgroups.
 - The attributes for the text are provided in the attributes map.
 - The INDENTWEIGHT option is used to indent the values.
- The second YAXISTABLE statement is used to display the count + percent values on the left.
- The last YAXISTABLE statement is used to display three statistics on the right.
- In a wide graph like this one, it helps to use the horizontal bands as a visual aid.
 - The REFLINE statement is used to display the bands using the “ref” variable.
 - The value from the “ObsId” column is copied into the “Ref” column wherever we need the bands.
 - Wide reference lines are drawn wherever the “ref” variable is not missing.
 - DISCRETETHICKNESS=1 is used to set the refine thickness equal to the midpoint spacing.
- The YAXISTABLE automatically displays the variable label above the column, split to fit the space.
- The data needed to display the “PCI Better” indicators on the x-axis is in the data set. TEXT statements are used to display these values.
- Note the use of AXISEXTENT=data to display the axis line only over the data extent of the Hazard Ratio data.
- No annotation is needed to create this graph.

USING THE XAXISTABLE TO CREATE PLOT WITH A CUSTOM TIME AXIS

The features needed for the survival plot and the forest plots were instrumental in the design of the AXISTABLE statements. These graphs were often requested by users and building them using the traditional annotation process was very cumbersome and the process does not easily scale to different data. Axis table statements makes it easier to create such graphs. However, you can also use these statements in other creative ways to build custom graphs.

For the first example, let us review a common stock plot shown in Figure 14. To create this graph, I downloaded the historical stock data for Amazon for the last 6 months from the Nasdaq web site as a CSV file. I used a data step to clean the data and transform it into a format suitable for SAS, with a few additional columns that will be useful later. A portion of the data is shown in Figure 12.

date	high	low	open	close	datechar	dateref	move
27SEP17	\$955.30	\$943.30	\$948.00	\$950.87	27SEP17		Up
28SEP17	\$959.70	\$950.10	\$951.86	\$956.40	28SEP17		Up
29SEP17	\$964.83	\$958.37	\$960.11	\$961.35	29SEP17		Up
02OCT17	\$967.30	\$952.12	\$964.00	\$959.19	02OCT17	02OCT17	Down
03OCT17	\$963.69	\$950.37	\$958.00	\$957.10	03OCT17		Down
04OCT17	\$967.79	\$954.05	\$954.21	\$965.45	04OCT17		Up
05OCT17	\$981.51	\$969.64	\$970.00	\$980.85	05OCT17		Up
06OCT17	\$995.75	\$975.64	\$975.64	\$989.58	06OCT17		Up

Figure 12 – Stock data

Note in this data set, the original character date is retained. Also, at every start of a new month, the character date is copied into the “dateref” column. For other observations, this value is missing. Now, let us create a stock plot from this data using the “date” variable. The program is shown in Figure 13 and the resulting graph is shown in Figure 14.

```

title 'Stock Price for Amazon by Date';
proc sgplot data=amzn noborder;
  highlow x=date low=low high=high / open=open close=close
    lineattrs=(thickness=2) y2axis;
  y2axis display=(noline noticks nolabel) grid;
  xaxis display=(nolabel);
run;

```

Figure 13 – SGPLOT procedure code for the stock plot



Figure 14 – Stock plot

This graph uses a SAS date axis since the “Date” variable has a SAS date format. The values displayed in the graph are correctly positioned on a scaled date axis. This results in gaps where there is no data on a valid date, such as for weekends and holidays when the stock market is closed.

While the above graph is technically correct, this is not how industry standard stock charts are plotted. The standard practice in the industry is to plot the data in a continuous manner, where the values for Monday are shown immediately after Friday, without any gaps. The same is true for all holidays.

You can create such a graph by first sorting the data by ascending “datechar” variable, and then plotting the data with a discrete x-axis, so that all the values are shown equally spaced on the axis. This will achieve our goal except the x-axis will display all the date value at a 45° angle, creating a very cluttered axis (not shown).

The trick here is to suppress the x-axis ticks and values, and display only the date values we want using the XAXISTABLE. We also use the REFLINE statement to draw a vertical reference line at each displayed value. The code is shown in Figure 15, and the resulting graph is shown in Figure 16.

```

title 'Stock Price for Amazon by Date';
proc sgplot data=amzn noborder noautolegend dattrmap=attrmap;
  reffline dateref / axis=x lineattrs=graphgridlines;
  highlow x=datechar low=low high=high / open=open close=close
    group=move lineattrs=(thickness=2) y2axis attrid=Move;
  xaxistable dateref / nolabel valueattrs=(weight=bold);
  y2axis display=(noline noticks nolabel) grid;
  xaxis type=discrete display=(nolabel novalues noticks)
    discreteorder=data;
run;

```

Figure 15 – SGPLOT procedure code for the stock plot



Figure 16 – Stock plot

The stock plot in Figure 16 shows the data in the industry preferred form, where the data for each day is displayed next to the previous day when the stock market is open. There are no gaps for weekends or holidays. We have achieved this result by using the XAXISTABLE to display the date values from the “dateref” column.

- In the data, the “dateref” column contains text string for only the first working day of each month.
- The stock data is sorted by ascending “datechar” and plotted on a discrete x-axis.
- The x-axis ticks, values and label are hidden.
- A XAXISTABLE is used to display the associated “dateref” values at the bottom of the graph.
- Reference lines are drawn at each non-missing dateref value using the GraphGridLines style.
- Note in this graph, the tick values are at uneven intervals, based on the number of observations per month. The bars for the stock values are colored green when the stock closes above previous close.

USING THE YAXISTABLE FOR CUSTOM GRAPHS

Sometimes there is a need to display a custom axis where the tick values on the axis need to be something different from the default. Often, you can do that by providing alternate tick values for the axis by using the VALUES option. If the values need to be modified further, you can use the VALUESDISPLAY option to provide the text strings for the specific values to be replaced.

The data for this example is shown in Figure 18 on the right. The “y2” column contains the value from the column y for every other observation. Also, a “group” column is computed that will be used later.

```
title "Default Y axis for Y=4*X~{unicode '00b2'x}";
proc sgplot data=trans noborder noautolegend;
  needle x=x y=y / displaybaseline=auto);
  yaxis grid display=(nolabel);
  xaxis display=(nolabel);
run;
```

Figure 17 – SGPLOT procedure code for $y=4x^2$ plot

x	y	y2	group
0	0	0	1
10	400	.	1
20	1600	1600	1
30	3600	.	1
40	6400	6400	1
50	10000	.	1
60	14400	14400	2

Figure 18 - Data

The output graph is shown in Figure 19. The default y-axis displays the tick values and grid lines at a regular interval.




Figure 19 – Default plot of $y=4*x^2$

Now, what if we want to display the y-axis tick values at custom location, also with highlighting of some tick values? As mentioned earlier, you can change the axis values displayed on the y-axis by using the VALUES option. Then, the individual value string can be replaced using the VALUEDISPLAY option. However, this requires us to populate these values into the procedure syntax. But, there is no way to change the color of some of the tick values through axis options.

The program in Figure 20 uses the YAXISTABLE to display the values on the left side from the “y” column, with COLORGROUP=group. This displays every value in the “y” column on the left and also displays the “14400” value in red. The output graph is shown in Figure 21.

```

title "Custom Y axis for Y=4*X~{unicode '00b2'x} with color";
proc sgplot data=trans noborder noautolegend;
  styleattrs datacontrastcolors=(black red);
  refline y / lineattrs=graphgridlines;
  needle x=x y=y / displaybaseline=auto lineattrs=(thickness=10);
  yaxistable y / location=outside position=left valueattrs=(size=8)
    nomissingchar nolabel colorgroup=group;
  yaxis display=(nolabel noticks novalues) offsetmin=0;
  xaxis display=(nolabel);
run;

```

Figure 20 – SGPLOT procedure code for $y=4*x^2$ plot




Figure 21 – Custom axis for plot of $y=4*x^2$

We could have used the “y2” variable in the YAXISTABLE statement to display alternate values instead of every one. If needed, you can plot the values at different locations, or even put text strings instead of numbers. The key benefit

is that the axis table allows you to control this from your data, instead of having to set up the options in the procedure syntax.

USING THE YAXISTABLE FOR SPARKLINE PANEL

Finally, let us see how you can build a graph to view inpatient claims over 6 to 48 months and also view the data as a spark line that can give you a quick visual of the trend. Such a graph was presented at the SESUG Conference 2017 in Cary by Rick Andrews in the presentation "Methods for creating Sparklines using SAS".

Rick used the traditional SAS/GRAPH procedures and annotation to create the graph. A similar graph can be created using the SGPanel procedure with the YAXISTABLE statements. The data for this graph is simulated by me using random functions and is shown in Figure 22.

Obs	Year	Months	Value	mid	m06	m12	m24	m36	m48
1	2006	6	\$111,443
2	2006	12	\$109,270
3	2006	24	\$110,595
4	2006	36	\$112,246
5	2006	48	\$111,583
6	2006	.	.	110757.86	111443.01	109269.55	110595.15	112246.18	111583.15
7	2007	6	\$121,086
8	2007	12	\$120,664
9	2007	24	\$119,578
10	2007	36	\$119,037
11	2007	48	\$119,646
12	2007	.	.	119850.71	121085.67	120664.09	119578.12	119037.32	119645.82

Figure 22 – Inpatient claim run-out data


This data set contains the claim runout values by year and period. For each year, I have also put the monthly values in separate columns such as “m06”, “m12” and so on. I have also computed an average of all these columns in the column “mid”.

Now, I can use the SGPanel program shown in Figure 23 to display the data as a panel by year, including the monthly values, and a “SparkLine” graph showing the trend over the monthly periods by Year. The output graph is shown in Figure 23.

```

title 'Inpatient Claim Run-out Patterns per Year';
title2 'Run-out periods: 6, 12, 24, 36, and 48 months';
proc sgpanel data=final;
  format value m06 m12 m24 m36 m48 dollar8.0;
  styleattrs wallcolor=cxe7ebf7;
  panelby year / layout=rowlattice onepanel uniscale=column
    noheader noborder spacing=10;
  series x=months y=value / markers
    markerattrs=(symbol=circlefilled size=5);
  rowaxistable year m06 m12 m24 m36 m48 / y=mid nomissingchar
    valueattrs=(size=8) position=left pad=10 labelpos=top;
  rowaxis display=none;
  colaxis display=none;
run;
```

Figure 23 – SGPanel procedure code for the “SparkLine” panel.

**Figure 24 – “SparkLine” Panel graph**

Here are the key features of this graph:

- We use the SGPPANEL procedure with PANELBY year and LAYOUT=RowLattice.
- We display the SERIES plot of runout value by month. The axes display is suppressed.
- We use the ROWAXISTABLE to display Year, and the 5 monthly values.
- Note, we use Y=mid option to set the y axis value to the mean of the yearly values.
 - If we use the default y=value, then the values in the table will not be evenly positioned.

CONCLUSION

With SAS 9.4, the SGPPLOT, SGPPANEL procedures and GTL support the new axis table feature. This allows you to easily include textual data in your graph, aligned with the x or y axis. These statements were initially motivated by the popular survival and forest plots. Now, you can use axis table statements to include such data without the need for annotation.

Additionally, you can also use the axis table statements to create custom graphs. The examples show how to use these for creating custom x or y-axes. You can also use these to create complex graphs such as the “SparkLine” panel.

REFERENCES

Matange, Sanjay 2016. *Clinical Graphs Using SAS®*. 1st ed. Cary, NC: SAS Institute.

Andrews, Rick, Hadden, Louise and Allison, Robert. 2017. “Methods for creating Sparklines using SAS.” *Proceedings of SESUG*, Cary, NC.

Matange, Sanjay, “Graphically Speaking.” <https://blogs.sas.com/content/graphicallyspeaking/2017/11/14/spark-table/>.

RECOMMENDED READING

- *Base SAS® Procedures Guide*
- *Clinical Graphs Using SAS®*

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Sanjay Matange

SAS Institute Inc.,
100 SAS Campus Dr.
Cary, NC 27513
Sanjay.Matange@sas.com
<http://blogs.sas.com/content/graphicallyspeaking>

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.